

On the computation of Siegel modular forms of degree 2 with Sage

Sho Takemori

Abstract

In this paper, we introduce a package of Sage [9] for the calculation of Siegel modular forms of degree 2.

1 Introduction

Sage [9] is a free and open software for various areas of mathematics. With Sage, we can compute many number theoretical objects including modular forms of one variable i.e. elliptic modular forms. But we cannot compute modular forms of several variables such as Siegel modular forms with built-in functions of Sage. The author wrote a package [10] for Siegel modular forms of degree two. In this paper, we introduce the package by computing Hecke eigenforms. This paper does not contain any new mathematical results.

2 Definitions

In this section, we recall the definition and related topics of Siegel modular forms.

2.1 Definition of Siegel modular forms of degree n

Let n be a positive integer and define the Siegel modular group of degree n by

$$\Gamma_n := \{g \in \mathrm{GL}_{2n}(\mathbb{Z}) \mid {}^t g w_n g = w_n\}.$$

Here $w_n = \begin{pmatrix} 0_n & -1_n \\ 1_n & 0_n \end{pmatrix}$. Note that $\Gamma_1 = \mathrm{SL}_2(\mathbb{Z})$. Define Siegel upper half space \mathfrak{H}_n by

$$\mathfrak{H}_n := \left\{ Z = X + iY \mid X, Y \in \mathrm{Sym}_n(\mathbb{R}), Y \text{ is positive definite} \right\}.$$

For a non-negative integer k , let $M_k(\Gamma_n)$ the the set of holomorphic functions F on \mathfrak{H}_n satisfying the following condition:

$$F((AZ + B)(CZ + D)^{-1}) = \det(CZ + D)^k F(Z), \quad \forall \begin{pmatrix} A & B \\ C & D \end{pmatrix} \in \Gamma_n.$$

If $n = 1$, we add the cusp condition. We call an element of $M_k(\Gamma_n)$ a Siegel modular form of degree n and weight k (and level 1). If $n = 1$, $M_k(\Gamma_1)$ is equal to the space of elliptic modular forms of weight k . It is known that $M_k(\Gamma_n)$ is a finite dimensional vector space over \mathbb{C} .

2.2 Fourier expansion of Siegel modular forms of degree two

Let $F \in M_k(\Gamma_2)$ be a Siegel modular form of degree 2. We put $Z = \begin{pmatrix} \tau & z \\ z & \omega \end{pmatrix} \in \mathfrak{H}_2$. Then F has the following Fourier expansion:

$$F \left(\begin{pmatrix} \tau & z \\ z & \omega \end{pmatrix} \right) = \sum_{\substack{n, r, m \in \mathbb{Z} \\ n, m, 4nm - r^2 \geq 0}} a((n, r, m), F) \mathbf{e}(n\tau + rz + m\omega),$$

where $\mathbf{e}(z) = e(2\pi iz)$ for $z \in \mathbb{C}$. With the notation above, we define the Siegel operator $\Phi : M_k(\Gamma_2) \rightarrow M_k(\Gamma_1)$ by

$$\Phi(F) := \sum_{n=0}^{\infty} a((n, 0, 0), F) \mathbf{e}(nz).$$

We define the space of cusp forms $S_k(\Gamma_2)$ of degree 2 by

$$S_k(\Gamma_2) := \ker \Phi \subseteq M_k(\Gamma_2).$$

2.3 Hecke polynomials

Let $n = 1$ or 2 . For $m \in \mathbb{Z}_{\geq 1}$, let $T(m) \in \text{End}_{\mathbb{C}}(M_k(\Gamma_n))$ the m th Hecke operator. We omit the definition of $T(m)$. See [2] for the definition. For a prime p and $F \in M_k(\Gamma_n)$, define a polynomial $Q_p^{(n)}(F; X)$ as follows.

1. If $n = 1$, then we define

$$Q_p^{(1)}(F; X) = 1 - \lambda(p)X + p^{k-1}X^2.$$

2. If $n = 2$, then we define

$$\begin{aligned} Q_p^{(2)}(F; X) &= 1 - \lambda(p)X \\ &+ \left(\lambda(p)^2 - \lambda(p^2) - p^{2k-4} \right) X^2 - \lambda(p)p^{2k-3}X^3 + p^{4k-6}X^4. \end{aligned}$$

Remark 1. $Q_p^{(n)}(F; p^{-s})^{-1}$ is the Euler factor of spinor L -function of F .

3 Structure theorem for the ring of Siegel modular forms of degree 2

In this section, we recall the structure theorem for the ring of Siegel modular forms of degree 2 proved by Igusa [4]. The structure theorem and the explicit formula for Siegel Eisenstein series of degree 2 enable us to compute Siegel modular forms of degree 2 explicitly.

Let

$$M(\Gamma_2) = \bigoplus_{k \in \mathbb{Z}_{\geq 0}} M_k(\Gamma_2)$$

be the ring of Siegel modular forms of degree 2. Put

$$\begin{aligned} x_{10} &:= E_4E_6 - E_{10}, \\ x_{12} &:= 3^2 \cdot 7^2 E_4^3 + 2 \cdot 5^3 E_6^2 - 691E_{12}, \end{aligned}$$

where E_k is the Siegel-Eisenstein series of degree 2 and weight k . Then x_{10} and x_{12} are Siegel cusp forms of weight 10 and 12 respectively. For $k = 10, 12$, we put

$$X_k := \frac{1}{a((1, 1, 1), x_k)} x_k.$$

The following theorem was proved by Igusa [4].

Theorem 1. 1. *There exists a weight 35 cusp form X_{35} (we normalize X_{35} so that $a((2, -1, 3), X_{35}) = 1$).*

2. *E_4, E_6, X_{10}, X_{12} and X_{35} generate $M(\Gamma_2)$ as a \mathbb{C} -algebra.*

3. *E_4, E_6, X_{10} and X_{12} are algebraically independent over \mathbb{C} .*

The Fourier coefficients of Siegel-Eisenstein series of degree 2 was known by Kaufhold [5]. Aoki and Ibukiyama [3] proved that cusp form X_{35} of weight 35 can be written by a polynomial of Siegel-Eisenstein and its differentials. Thus the generators of the ring $M(\Gamma_2)$ can be written by the polynomials of Siegel-Eisenstein series of degree 2 and its differentials. Therefore we can compute the Fourier coefficients of an element of $M(\Gamma_2)$ explicitly.

4 Computation of elliptic modular forms

In this section, we compute Hecke polynomial of elliptic cusp forms by using built-in functions of Sage.

```
R.<x> = PolynomialRing(QQ, 1, order='neglex')
def euler_factor_of_1(f, p):
    wt = f.weight()
    return 1 - f[p]/f[1]*x + p^(wt-1)*x^2
wts_of_one_dim = \
    [k for k in range(12, 30)
     if CuspForms(1, k).dimension() == 1]
```

In the code above, we compute $Q_p^{(1)}(f; X)$ for $p = 2$ and $f \in S_k(\Gamma_1)$ with $\dim S_k(\Gamma_1) = 1$. The function `euler_factor_of_1` takes an eigenform and a prime p and returns $Q_p^{(1)}(f; X)$. `wts_of_one_dim` is the list of the positive integers k such that $12 \leq k < 30$ and $\dim S_k(\Gamma_1) = 1$.

```
euler_factor_at_2 = {}
for k in wts_of_one_dim:
    f = CuspForms(1, k).basis()[0]
    euler_factor_at_2[k] = euler_factor_of_1(f, 2)
```

The python's dictionary `euler_factor_at_2` is a dictionary such that $k \mapsto Q_2^{(1)}(f_k; X)$, which value is as follows:

```
sage: euler_factor_at_2
{12: 1 + 24*x + 2048*x^2,
 16: 1 - 216*x + 32768*x^2,
 18: 1 + 528*x + 131072*x^2,
 20: 1 - 456*x + 524288*x^2,
 22: 1 + 288*x + 2097152*x^2,
 26: 1 + 48*x + 33554432*x^2}
```

For example the $Q_2^{(1)}(\Delta; X) = 1 + 24X + 2048X^2$, where Δ is the Ramanujan's delta.

5 Computation of Siegel modular forms in Sage

In this section, we compute Siegel modular forms of degree 2 by using the package [10]. The following code has been tested under Sage 6.11 and “degree2” (revision 706bfe).

5.1 Computation of generators of $M(\Gamma_2)$

The generator X_{10} can be obtained by the function `x10_with_prec(prec)`. Here the argument `prec` is a positive integer and this function computes the Fourier coefficients of X_{10} for

$$\{(n, r, m) \mid 0 \leq n, m \leq \text{prec}, 4nm - r^2 \geq 0\}.$$

X_{12} and X_{35} can be obtained by the function `x12_with_prec(prec)` and `x35_with_prec(prec)`. Siegel-Eisenstein series E_k can be obtained by the function `eisenstein_series_degree2(k, prec)`. Here are examples.

```
from degree2.all import *

prec = 4
# The cusp forms of weight 10 and 12.
X10 = x10_with_prec(prec)
X12 = x12_with_prec(prec)
```

```
# Fourier coefficient of X10 at (1, 1, 1).
X10[(1, 1, 1)] # => 1
# Fourier coefficient of X10 at (3, 5, 4).
X10[(3, 5, 4)] # => 2736
```

5.2 Computation of Hecke polynomials

We calculate $Q_p^{(2)}(F; X)$ for $p = 2$, $F \in S_k(\Gamma_2)$ and a small weight k . For $F = X_{10}$ and $p = 2$, we can compute $Q_p^{(2)}(F; X)$ as follows:

```
sage: X10 = x10_with_prec(4)
sage: X10.euler_factor_of_spiner_1(2).factor()
(-1 + 256*x) * (-1 + 512*x) * (1 + 528*x + 131072*x^2)
```

The last factor is equal to $Q_2^{(1)}(f_{18}, x)$, where $f_{18} \in S_{18}(\Gamma_1)$ is an eigenform. Thus we have

$$Q_2^{(2)}(X_{10}; X) = (1 - 2^8 X)(1 - 2^9 X)Q_2^{(1)}(f_{18}, X).$$

Here is another example.

```
sage: Q12 = X12.euler_factor_of_spiner_1(2)
sage: Q12.factor()
(-1 + 1024*x) * (-1 + 2048*x) * (1 + 288*x + 2097152*x^2)
```

Thus we also have

$$Q_2^{(2)}(X_{12}; X) = (1 - 2^{10} X)(1 - 2^{11} X)Q_2^{(1)}(f_{22}, X).$$

But not every eigenform of $S_k(\Gamma_2)$ is related to an eigenform of $S_k(\Gamma_1)$. We compute cuspidal eigenform $X_{20} \in S_{20}(\Gamma_2)$ whose Hecke eigenvalue of $T(2)$ is equal to -840960 . The cusp form X_{20} is not related to elliptic modular forms.

```
# S20 is the space of cusp forms of degree 2 and weight 20.
sage: S20 = CuspFormsDegree2(20, prec = 4)
sage: S20.hecke_charpoly(2).factor()
# X20 is an eigenform of weight 20 whose eigenvalue of
# T(2) is -840960.
sage: X20 = S20.eigenform_with_eigenvalue_t2(-840960)
```

```
sage: X20.euler_factor_of_spinor_1(2).factor()
1 + 840960*x + 390238044160*x^2 + 115580662311813120*x^3
+ 18889465931478580854784*x^4
```

5.3 Maass relation and Saito-Kurokawa lift

In this subsection, we explain our examples above by the theorem proved by Maass, Andrianov and Zagier.

Before we state the theorem, we introduce the Maass relation and Maass subspace.

Definition 1 (Maass relation). For a cusp form $F \in S_k(\Gamma_2)$, we consider the following condition.

$$a(n, r, m) = \sum_{d>0, d|\gcd(n,r,m)} d^{k-1} a(1, r/d, mn/d^2), \quad (5.1)$$

for all $n, m, 4nm - r^2 \geq 0$. Here we put

$$a(n, r, m) = a((n, r, m), F).$$

We denote by $S_k^*(\Gamma_2)$ the set of Siegel cusp forms $F \in S_k(\Gamma_2)$ satisfying the condition above. We call $S_k^*(\Gamma_2)$ the Maass subspace.

For $F \in M_k(\Gamma_2)$ and (n, r, m) , we can check the equation (5.1) by the method `satisfies_maass_relation_for`.

```
sage: X10.satisfies_maass_relation_for(2, 1, 2)
True
sage: X12.satisfies_maass_relation_for(2, 1, 2)
True
sage: X20.satisfies_maass_relation_for(2, 1, 2)
False
```

The following theorem was proved by Maass [6] [7] [8], Andrianov [1] and Zagier [11]. And this theorem explains the relation between X_{10} (resp. X_{12}) and f_{18} (resp. f_{22}).

Theorem 2. *Let k be an even number. The Maass subspace $S_k^*(\Gamma_2)$ is stable under the action of Hecke operators. There exists a one-to-one correspondence between an eigenform $F \in S_k^*(\Gamma_2)$ and an eigenform $f \in S_{2k-2}(\Gamma_1)$ given by*

$$Q_p^{(2)}(F; X) = (1 - p^{k-2}X)(1 - p^{k-1}X)Q_p^{(1)}(f; X).$$

Remark 2. The existence of the lift $f \mapsto F$ was conjectured by H. Saito and Kurokawa independently.

Sho Takemori
 Department of Mathematics,
 Kyoto University
 Kitashirakawa-Oiwake-Cho, Sakyo-Ku,
 Kyoto, 606-8502, Japan
 E-mail: takemori@math.kyoto-u.ac.jp

References

- [1] A. N. Andrianov, *Modular descent and the Saito-Kurokawa conjecture*, *Inventiones mathematicae* **53** (1979), no. 3, 267–280.
- [2] A. N. Andrianov and V. G. Zhuravlev, *Modular forms and Hecke operators*, no. 145, American Mathematical Soc., 1995.
- [3] H. Aoki and T. Ibukiyama, *Simple graded rings of Siegel modular forms, differential operators and Borcherds products*, *International Journal of Mathematics* **16** (2005), no. 03, 249–279.
- [4] J. Igusa, *On siegel modular forms of genus two*, *American Journal of Mathematics* (1962), 175–200.
- [5] G. Kaufhold, *Dirichletsche Reihe mit Funktionalgleichung in der Theorie der Modulfunktion 2. Grades*, *Mathematische Annalen* **137** (1959), no. 5, 454–476.
- [6] H. Maass, *Über eine Spezialschar von Modulformen zweiten Grades*, *Inventiones mathematicae* **52** (1979), no. 1, 95–104.
- [7] ———, *Über eine Spezialschar von Modulformen zweiten Grades (II)*, *Inventiones mathematicae* **53** (1979), no. 3, 249–253.
- [8] ———, *Über eine Spezialschar von Modulformen zweiten Grades (III)*, *Inventiones mathematicae* **53** (1979), no. 3, 255–265.
- [9] W. A. Stein et al., *Sage Mathematics Software (Version 6.1.1)*, The Sage Development Team, 2014, <http://www.sagemath.org>.

- [10] S. Takemori, *degree2*, <https://github.com/stakemori/degree2>.
- [11] D. Zagier, *Sur la conjecture de Saito-Kurokawa (d'apres H. Maass)*, Seminar on Number Theory, Paris, vol. 80, 1979, pp. 371–394.